What is DBCC CHECKIDENT current column value
Not sure what you mean by "recovered" - current column value tells you the maximum value in the table. It's meant to show you when the maximum value is greater than the value SQL Server assigned last (and from which it determines what it thinks it's going to assign next, even if in practice this isn't always true). Example:

SET NOCOUNT ON;
CREATE TABLE dbo.smudge(id INT IDENTITY);
GO
-- insert 11 rows
INSERT dbo.smudge DEFAULT VALUES;
GO 11
DBCC CHECKIDENT('dbo.smudge'); -- 11, 11
GO
-- delete the largest id
DELETE dbo.smudge WHERE id = 11;
GO
DBCC CHECKIDENT('dbo.smudge'); -- 11, 10
GO
-- reseed the table
DBCC CHECKIDENT('dbo.smudge', RESEED, 1);
GO
DBCC CHECKIDENT('dbo.smudge'); -- 1, 10
GO
-- insert another row
INSERT dbo.smudge DEFAULT VALUES;
GO
DBCC CHECKIDENT('dbo.smudge'); -- 11, 11
GO
-- insert a much higher row
SET IDENTITY_INSERT dbo.smudge ON;
INSERT dbo.smudge(id) VALUES(255);
SET IDENTITY_INSERT dbo.smudge OFF;
GO
DBCC CHECKIDENT('dbo.smudge'); -- 255, 255
GO
DROP TABLE dbo.smudge;
GO